

# 「EQUATRAN-M」技術計算用連立方程式解法言語

(1)

## EQUATRAN-M はパソコン技術計算の世界を拡げる

佐渡友秀夫\*

宮原星中\*\*

今回から始まる連載でパソコン用に開発された技術計算用問題解法言語「EQUATRAN-M」(EQUAtion TRANslator for Microcomputer)の紹介をする。

EQUATRAN-Mは方程式の表現に便利ないくつかのきまり(文法)をもった言語であるが、パソコン(現在のところ16ビットCPU)用のソフトウェアの名前として読み進んでもらった方がわかり易い。EQUATRAN-Mは与えられた連立方程式を変形し、計算方法と手順(アルゴリズム)を発生、必要な入力値・初期値を要求して数値計算を実行し、答を表示または印刷する機能をもっている。~~~~~の部分をコード化するのがプログラミングといわれ、多くの方がBASICとかFORTRANで苦労した経験をお持ちだと思う。EQUATRAN-Mによれば、技術者や研究者は解こうとする問題を定式化して入力するだけでプログラミングせずに解を得ることができるので、EQUATRAN-Mは業務の効率化のための強力なツールとなる。

本連載では、まず2回にわたってEQUATRAN-Mの機能の概要と規則を説明し、次いで化学工業の諸分野で現われる実例をできるだけ多く取り上げて使用法を詳細に解説、最後に知っておくと便利なことなどをまとめて参考に供する予定である。

### 1. なぜ、いま EQUATRAN-M か

マイコンがパソコンとなりOAブームをひき起

してから5年以上たつが、ここ2、3年の16ビットパソコンの出現により本格的なビジネスユースが始まっているといえよう。それは、16ビット機がコストに比べて抜群の高性能を提供しているからで、これをいかに業務に活用するか各社必死になるのも当然である。しかし、8ビット機時代に「コンピュータ、ソフトなければただの箱」と喝破された反省から、業務上の使用に耐えうる汎用ソフトウェア(簡易言語)が登場し、果した役割は非常に大きい。つまり、業務に合せてハードウェアを選びソフトを開発するという従来のオフィスコンピュータの常識から、市販の汎用ソフトの機能に業務の方を合せそのソフトがのるハード(パソコン)の機種を選んで利用する方が有利であるとの発想の転換を促した。その結果、事務計算、管理計算などオフィス業務の合理化のため簡易言語は爆発的に普及した。

一方、技術計算の世界では適用分野が多岐にわたること、事務計算に比べて担当者のプログラミング能力が高いことがかえって汎用な技術計算用ソフトの発達を阻害し、業務発生の度毎に不便を感じながらもプログラムを作つて何とかこなしているのが実状であろう。今後、パソコンは単体としての性能の向上とともに、ホストコンピュータとの間でデータのやりとりをする多機能端末として、さらに、独立した解析・設計業務の遂行能力を併せもつエンジニアリングワークステーションとして位置づけられていくであろうから、技術計算分野でも専門家を面倒なプログラミング作業から解放し、より高度な解析を可能ならしめるツールキットの出現が待たれていたといえよう。

このような状況下で、あらゆる分野の技術計算

\* Hideo Sadotomo 三井東庄化学(株) システム部 主席部員

\*\* Koreatsu Miyahara 三井東庄化学(株) システム部 次長

に共通な問題解決言語として登場した EQUATRAN-M は、世界的に見ても類のない高水準な機能をもっており、時代の要請に十分応える性能を発揮するであろう。

ここで、EQUATRAN-M の生立ちについて簡単に触れておく。

三井東圧化学㈱では、昭和30年代から大型コンピュータを導入して技術計算を実施しており<sup>1~6)</sup>秀れたパッケージプログラム群を開発して使用している<sup>7,8)</sup>。その中の一つに EQUATRAN がある。化学プロセスの物質収支計算とか熱交換器等の機器の設計や運転解析をする際に、プロセスの要求仕様や測定項目が変ると解くべき方程式は同じでも計算手順が一変することが多く、その度にプログラムを書き直し検査するので大変な労力を要する。そこで、解析作業を効率化し現場への即応性や業務移管の容易さを達成する方法を模索・検討し、従来の手続き型プログラミングを超える言語の発想が生まれた。この発想を具現化するシステムを大型コンピュータ用に PL/1 で開発を進め、昭和50年に完成した<sup>9)</sup>。これは50年代の省エネルギーのためのプロセス合理化検討に大いに利用され、機能の拡充と性能の強化が行われてきたが、16ビットパソコンの出現を待つて移植作業が着手された。EQUATRAN-M の熟成した機能とパソコンの最大の魅力である使い勝手の良さが十分に発揮されるように配慮しつつ仕様が決定され、C 言語で書き改められて EQUATRAN-M としてお目見えした次第である<sup>10)</sup>。

化学工学の分野から誕生した EQUATRAN-M であるが、連立方程式を解くという本質はこの分野だけに留まらず、機械装置の設計、強度計算、電気回路の設計、栄養配分、調剤、計量経済などあらゆる領域で方程式を扱う業務に適用できるものである。

## 2. EQUATRAN-M とは

まず、簡単な例題について BASIC によるプログラミングと比較しながら EQUATRAN-M の特徴を説明する。

### 〈例題 1.1：混合液の沸点計算〉

ベンゼン、トルエン、p-キシレン 3 成分混合液の沸点  $t$  [°C] と気相のモル分率  $y$  を求めよ。

混合液のモル分率  $x$  は (0.5, 0.3, 0.2) であり、各成分の蒸気圧  $P^\circ$  [mmHg] は Antoine の式(1-1) で表わされ、その定数は表1-1 で与えられる<sup>11)</sup>。

$$\log_{10} P^\circ = a - b/(c + t) \quad (1-1)$$

表 1-1 Antoine 式の定数

	ベンゼン	トルエン	p-キシレン
a	6.90565	6.95464	6.99052
b	1211.03	1344.80	1453.43
c	220.790	219.482	215.307

解) この混合液を理想溶液として扱うと、第  $i$  成分の  $x$  と  $y$  の関係は Raoult の法則より全圧  $P$  [mmHg] と  $P_i^\circ$  を用いて次式で表わされる。

$$P y_i = P_i^\circ x_i \quad (i=1, 2, 3) \quad (1-2)$$

$y$  はモル分率であるから

$$\sum_{i=1}^3 y_i = 1 \quad (1-3)$$

これを解くのに、 $t$  を与えて  $P$  を求めるのは電卓でも容易にできるが、逆に  $P$  を与えて  $t$  を求めるためには  $t$  に関する繰り返し計算を必要とする。この問題を EQUATRAN-M の記法で書くと表 1-2 になる注)。

表 1-2 を見てわかるように解こうとする式がそのまま書かれており、式を書きやすいと同時にこのリストがドキュメントないし報告書の役を果す。表 1-2 の内容をキーインし、「RUN」というコマンドを与えてエンターキー（以下回と書く）を押すと、EQUATRAN-M は方程式を展開し、 $t$  に初期値を与えて繰り返し計算をして解くという手順を組立てる。そして、表 1-3 で示すように  $P$  の入力値と  $t$  の初期値を 1 つづつ対話方式で要求してくれる。このとき、表 1-2 の第 1 行が自動的に

注) /\* と \*/ ではさまれた部分（例えば、1行目や 3 ~ 7 行目）は注釈である。変数名については大文字と小文字は区別されるが、予約語やコマンド（命令）ではどちらでも同じである。加減乗除 (+-\*/>) とべき乗 (^) の記号と演算優先順位は BASIC と全く同一である。左端の数字は行番号を示しているが、9 行目の VAR とは  $P^\circ$ ,  $x$ ,  $y$ ,  $a$ ,  $b$ ,  $c$  に 3 成分に対応する内容をもたせるための宣言文である。従って、11 行目の表現は 3 つの成分に関する方程式をまとめたものと解釈される。14 ~ 16 行の右辺の ( ) 内は配列定数を表わしている。19 行目の SUM ( ) は (1-3) 式の  $\Sigma$  に相当する組込み関数である。最後の行は  $P$  を入力して、計算結果として得られる  $t$  と  $y$  の値を表示することを指示している。

表 1-2 例題 1.1 のリスト

```

1: /* 沸点の計算 */
2:
3: /* P : 全圧 [mmHg]
4:   PG : 分圧 [mmHg]
5:   t : 温度 [°C]
6:   x : 液相モル分率 [-]
7:   y : 気相モル分率 [-] */
8:
9: VAR P0(3), X(3), Y(3), a(3), b(3), c(3)
10:
11: log10(P0) = a - b/(c + t) /* Antoine 式 */
12:
13: /* benzene toluene p-xylene */
14: a = ( 6.90565, 6.95464, 6.99052 )
15: b = ( 1211.03, 1344.80, 1453.43 )
16: c = ( 220.790, 219.482, 215.307 )
17:
18: P + y = P0 * x
19: SUM(y) = 1
20: x = ( 0.5, 0.3, 0.2 )
21:
22: INPUT P; OUTPUT t, y

```

表 1-3 入力値と初期値

```

/* 沸点の計算 */
[ 入力値(=) または 初期値(#) ]
P =? 760
t =? 50

```

表 1-4 計算結果

```

[ 計算結果 ]
t = 94.458679
y =
1) 0.762605      2) 0.18507      3) 0.052325

```

見出しとして表示される。そこで、しかるべき値（この場合  $P$  に 760 を与え、 $t$  に 50 を仮定）をキーインし回を押すと計算の実行に入り、計算結果として表 1-4 のように  $t$  と  $y$  の値を標準形式で表示してくれる（出力の様式に気を使わなくてよい）。これで答が得られた。

一方、この問題を BASIC で書くと表 1-5 のようになろう注)。温度  $T$  を仮定してその温度での  $\sum_{i=1}^3 y_i$  を計算し、結果が 1 になるよう Newton 法でくり返し計算をしている。つまり、(1-3) 式を変形して、

$$f(T) = \sum_{i=1}^3 y_i - 1 \quad (1-4)$$

とし（プログラム中で SUM は  $\sum_{i=1}^3 y_i$  を意味している）、 $f(T)=0$  となるように

$$\text{新 } T = T - f(T)/\{\partial f(T)/\partial T\} \quad (1-5)$$

で  $T$  を修正する。 $\partial f(T)/\partial T$  は次式で求める。

$$\partial f(T)/\partial T = \sum_{i=1}^3 (\partial y_i / \partial T)$$

注) BASIC では大文字と小文字の使い分けはできない

表 1-5 例題 1.1 の BASIC プログラム

```

100 ' 沸点の計算
110 DIM P0(3), X(3), Y(3), A(3), B(3), C(3)
120 A(1) = 6.90565 : A(2) = 6.95464 : A(3) = 6.99052
130 B(1) = 1211.03 : B(2) = 1344.8 : B(3) = 1453.43
140 C(1) = 220.79 : C(2) = 219.482 : C(3) = 215.307
150 X(1) = .5 : X(2) = .3 : X(3) = .2
160 CLS : PRINT "沸点の計算"
170 INPUT "P ="; P : INPUT "t ="; T
180 EPS = .0001
190 ITR = 0
200 ' iteration entry point
210 ITR = ITR + 1
220 IF ITR > 10 THEN PRINT "not converged": GOTO 380
230 SUM = 0
240 DF = 0
250 FOR I=1 TO 3
260 P0(I) = 10^(A(I)-B(I)/(C(I)+T))
270 Y(I) = P0(I)*X(I)/P
280 SUM = SUM + Y(I)
290 DF = DF + (2.303*B(I)/(C(I)+T)^2)*Y(I)
300 NEXT I
320 F = SUM - 1
330 IF ABS(F) > EPS THEN T = T-F/DF : GOTO 200
340 ' iteration converged
350 PRINT : PRINT "[ 計算結果 ]"
360 PRINT "t ="; T
370 PRINT "y ="; Y(1); Y(2); Y(3)
380 END

```

表 1-6 入力値

```

沸点の計算
P =? 760
t =? 50

```

表 1-7 計算結果

```

[ 計算結果 ]
t = 94.4587
y = .762606 .18507 .0523247
Ok

```

$$= \sum_{i=1}^3 \left\{ 2.303 b_i / (c_i + T)^2 \right\} \cdot y_i \quad (1-6)$$

もちろん、 $T$  の修正のために Newton 法以外の手法も考えられるが、いずれにしても BASIC でプログラムを作るには文法の他に計算手順についても相当な知識と経験が必要である（これは FORTRAN でも同様である）。さて、このプログラムを ‘RUN’ 回として実行させると表 1-6 のように入力を要求してくるので  $P$  に 760、 $T$  に 50 をキーインすると表 1-7 のように計算結果の  $T$  と  $Y$  が表示される。

このように連立方程式を解く上で EQUATRA N-M がいかに便利かを示したが、BASIC プログラムとの基本的な違いは「=」の役割にある。BASIC での「=」は右辺の式の値を左辺の変数に代入することを指示しているが、EQUATRA N-M では右辺の式と左辺の式が等しいこと（つまり数学的等号）を意味している。従って、BA

SIC では右辺に未知変数がきてはならないので方程式の变形とそれらを並べる順が極めて重要であるが、EQUATRAN-M ではそのようなことを一向に気にする必要はない。方程式や条件を自然な形で任意の順序に書いていければよい。いかに解くかは EQUATRAN-M にお任せである。だから、いったん EQUATRAN-M で書いた数式モデルの一部の方程式を修正することも、条件として値が与えられていた変数を逆に計算結果として求めることも自在にできる。例えば、

### <例題 1.2：露点計算>

例題1.1の条件で  $y$  が (0.6, 0.3, 0.1) のとき露点  $t$  と  $x$  を求めよ。

解) 例題1.1の沸点計算と違うのは  $x$  の代りに  $y$  が与えられていることで、従って(1-3)式に代って

$$\sum_{i=1}^3 x_i = 1 \quad (1-7)$$

が使われる。EQUATRAN-Mによるリストと計算結果を表1-8に示す。表1-2の19, 20, 22行目（と1行目）を書き直せばよい。

まことに簡単明瞭であるが、これを BASIC で書くとどうなるかは読者にお試し願いたい。

このように、EQUATRAN-Mを使えば、技術者、研究者が今まで方程式を解くことに割いていた貴重な能力と時間を省くことができ、問題を解決するという専門家本来の業務に専念できる。また、教育の場においても解法の訓練から与えられた文章題を解釈し、定式化して、得られた結果について考察するという演習に重点を置くことが可能となり、創造性と応用力を養うための知的ツールとして位置づけられよう。

## 3. EQUATRAN-M で何ができるか

それでは、基本的な機能について引続き例題を用いて説明する。

### 3.1 連立方程式の求解

EQUATRAN-M は一次方程式、高次代数方程式、初等関数（三角関数、対数関数、指數関数など）を含む方程式およびそれらが組合された連立方程式の解を自動的に求めることを最大の特徴としている。

表 1-8 例題 1.2 の解

```

1: /* 露点の計算 */
2:
3: /* P : 全圧 [mmHg]
4:   P0: 分圧 [mmHg]
5:   t : 溫度 [°C]
6:   x : 液相モル分率 [-]
7:   y : 気相モル分率 [-] */
8:
9: VAR P0(3), x(3), y(3), a(3), b(3), c(3)
10:
11: log10(P0) = a - b/(c + t) /* Antoine 式 */
12:
13: /* benzene toluene p-xylene */
14: a = ( 6.90565, 6.95464, 6.99052 )
15: b = ( 1211.03, 1344.80, 1453.43 )
16: c = ( 220.790, 219.482, 215.307 )
17:
18: P * y = P0 * x
19: SUM( x ) = 1
20: y = ( 0.6, 0.3, 0.1 )
21:
22: INPUT P; OUTPUT t, x

/* 露点の計算 */
[ 入力値(=) または 初期値(#) ]
P = 760
t = # 120

[ 計算結果 ]
t = 102.042748
x =
1) 0.319389 2) 0.385359 3) 0.295252

```

### <例題 1.3：水蒸気の定圧分子熱式の決定>

気体の定圧分子熱  $C_p$  [kcal/kg-mol·deg] は温度  $T$ [°K] の関数として次式でよく近似できることが知られている<sup>12)</sup>。水蒸気について表1-9の実測値からこの定数を決定せよ。

$$C_p = a + bT + cT^2 + dT^3 \quad (1-8)$$

表 1-9 水蒸気の定圧分子熱<sup>13)</sup>

$t$ [°C]	100	140	180	250
$C_p$	2.017	1.984	1.971	1.980

解) (1-8) 式は  $a$ ,  $b$ ,  $c$ ,  $d$  の 4 つの未知数を含むから 4 組の実測値を温度の単位変換をして代入し、4 元連立一次方程式を解くことになる。リストと計算結果を表1-10に示す。

表 1-10 例題 1.3 の解

```

1: /* 水蒸気の定圧分子熱式の決定 */
2:
3: /* Cp = a + b*T + c*T^2 + d*T^3
4:   Cp : 分子熱 [kcal/kg-mol·deg]
5:   T : 溫度 [°K]
6:   Data from I.C.T. No.5 p.82 */
7:
8: 2.017 = a + b*373.15 + c*373.15^2 + d*373.15^3
9: 1.984 = a + b*413.15 + c*413.15^2 + d*413.15^3
10: 1.971 = a + b*453.15 + c*453.15^2 + d*453.15^3
11: 1.980 = a + b*523.15 + c*523.15^2 + d*523.15^3
12:
13: OUTPUT a, b, c, d

/* 水蒸気の定圧分子熱式の決定 */
[ 計算結果 ]
a = 4.276842
b = -0.012977
c = 2.382229E-005
d = -1.417749E-008

```

(1-8) 式に実測値を代入して書き並べただけである。8~11 行で左辺と右辺が入れ替っていても

表 1-11 組込み関数一覧表

「EQUATRAN-M」の表記	数学的表記	内 容
exp(x)	$e^x$	指数
exp10(x)	$10^x$	10のべき乗
sqre(x)	$x^2$	平方
loge(x)	$\log_e(x)$	自然対数
log10(x)	$\log_{10}(x)$	常用対数
sqrt(x)	$\sqrt{x}$	平方根
sin(x)	$\sin x$	正弦
cos(x)	$\cos x$	余弦
tan(x)	$\tan x$	正接
asin(x)	$\sin^{-1}x$	逆正弦
acos(x)	$\cos^{-1}x$	逆余弦
atan(x)	$\tan^{-1}x$	逆正接
sinh(x)	$\sinh x$	双曲線正弦
cosh(x)	$\cosh x$	双曲線余弦
tanh(x)	$\tanh x$	双曲線正接
asinh(x)	$\sinh^{-1}x$	双曲線逆正弦
acosh(x)	$\cosh^{-1}x$	双曲線逆余弦
atanh(x)	$\tanh^{-1}x$	双曲線逆正接
sum(y)	$\sum_{i=1}^n y_i$	配列変数の総和
prod(y)	$\prod_{i=1}^n y_i$	配列変数の総積
max(x, y)	—	大きな方の値
min(x, y)	—	小さな方の値
if(x)	—	論理値化
abs(x)	$ x $	絶対値

構わない。

ところで、EQUATRAN-Mには初等関数の他に総和、総積、絶対値などを与える組込み関数が用意されているが、その一覧表を表1-11として掲げる(小文字でも大文字でもよい)。これらを用いれば大概の方程式を記述するのに事欠かない。

通常、化学工学の分野で遭遇する問題は、多くの変数間に一次方程式で表わされる線形な関係があり、その他にいくつかの非線形な方程式(くり返し計算を必要とする)が混在する場合が多い。例題1.1もそうであるが、プロセスの物質収支・熱収支をとる問題などまさにその典型で、しかも成分の数とストリームの数を掛けた数の未知変数を扱わねばならず、すぐに100変数にも達する。EQUATRAN-Mはそのような大きな問題を解くのにも大変適している。ただし、数値的に解を求めてるので、複数の解をもつ方程式の全ての解を同時に求めることはできない。その場合は初期値を変えて再試行することになる。とはいって、物理量を対象とする実際の問題では変数の範囲を規定されることが多く、そのときには唯一の解で十

分である。

EQUATRAN-Mは与えられた方程式群を展開し計算手順を組立ててオブジェクトコードを発生してくれるが、この過程で方程式の書き方の不備や未知変数と方程式の数の不一致などを診断し、必要に応じてメッセージを出す。このステップをコンパイルという。次に、計算手順に従い数値計算が実行され計算結果が得られるのだが、相当大型で複雑な問題をも確実に処理できるように、パソコンのメモリ、計算速度、計算精度、収束性を十分考慮した数値計算手法が用意されている。なお、演算は倍精度で実行される。このステップをゴーという。

ちなみに、対処できる問題の大きさはパソコンのメモリ容量(最小限 256k バイト)に依存するが、最小でも 100 変数は扱え、384k バイトあれば 400 変数まで可能である。また、処理に要する時間は NEC の PC9801F2 を使った場合で、少し大き目の問題として文献<sup>14)</sup>のプロセス物質収支計算(変数の数74、非線形方程式の数3)を例にとれば、コンパイルに15秒、ゴーは初期値によるが例えれば40秒である。この時間を数値演算プロセッサを付加すれば大幅に短縮できる。

### 3.2 最適化計算

EQUATRAN-Mのもう一つの特徴は、複数の不等号制約条件付きの多変数最適化計算ができることがある。最適化計算のアルゴリズムはなかなか厄介なもので、BASIC ではとてもプログラムを作る気が起きないが、EQUATRAN-Mでは連立方程式の求解と同様に関係する方程式と目的関数、制約条件を書き下してやれば、方程式を満足しきつ目的関数が最大もしくは最小となる変数の組を求めることができる。

#### <例題 1.4：2変数最適化問題>

(1-9) 式で与えられる目的関数を(1-10)式の制約条件下で最小にする  $x, y$  の値を求めよ。ただし、 $0 \leq x \leq 100, 0 \leq y \leq 100$  とする。

$$p = 3x^2 - 2xy + 3y^2 - 10x - 50y + 275 \quad (1-9)$$

$$x + y > 10 \quad (1-10)$$

表1-12にリストと計算結果を示す。最適化計算については次回に詳しく説明しよう。

### 3.3 方程式の表記法

表 1-12 例題 1.4 の解

```

1: /* 2 变数 最適化 問題 */
2:
3: p = 3*x^2 + 2*x*y + 3*y^2 + 10*x + 50*y + 275
4: c = x + y > 10
5:
6: FIND ( x # 10 [0,100], y # 5 [0,100] ) ..
7:      minimize p under c until 0.1%
8:
9: output p, x, y

/* 2 变数 最適化 問題 */
[ 入力値 (=) または 初期値 (#) .. ]
x          # 10
y          # 5

[ 計算結果 ]
p = 0.015437
x = 4.996364
y = 10.070439

```

これまで例示したように EQUATRAN-M の書式は自然で、方程式の並びは自由である。その上のような表現方法ができる。

- 式の長さは 2,000 文字まで可能で、2 行以上にわたってもよい（継続記号「..」で示す——表 1-12 の第 6 行）。あるいは、1 行に 2 つ以上の式があってもよい（「;」で区切る——表 1-2 の第 22 行）。なお、リストの行数は 600 行まで許されている。
- 変数名は英字で始まる 8 文字以内の英数字とアンダーラインで示し、変数名に限って英字の大文字と小文字は区別される。
- 配列変数（ベクトル、マトリクス）が使える。配列変数の演算は対応する要素毎の演算として定義される。例題 1.1 でもベクトル（一次元配列）を用いているが、例題 1.3 をベクトル表示すると表 1-13 のように非常に簡潔に表現できる。なお、表 1-13 の第 5 行のようにベクトルとスカラー（定数でも変数でもよい）の演算ではスカラーの次元が相手のベクトルの大きさに拡張されて解釈される。また、ベクトルの集まりをマトリクスとして表現できる。大きな問題を書き下すときはベクトル・マトリクス表現は必須であり、殊にプロセスフローを扱うときは大変便利である。
- 数表を引くことができる。2 つの変数の関係が数表で定義されるとき、表の間を相互に補間することができる。また、パラメータ変数を含む数表（二次元の表）も扱えるので、機器設計などに現われる線図を数式で近似する必要がなく数表として定義すればよい。
- 適用範囲によって異なる式を定義できる。

表 1-13 例題 1.3 の別解

```

1: /* 水蒸気の定圧分子熱式の決定 */
2:
3: VAR Cp(4), t(4), T(4)
4:
5: Cp = a + b*T + c*T^2 + d*T^3
6: /* Cp : 分子熱 [kcal/kg-mol/deg]
7: T : 温度 [°K] */
8:
9: /* Data from I.C.T. No.5 p.82 */
10: Cp = ( 2.017, 1.984, 1.971, 1.980 )
11: t = ( 100, 140, 180, 250 )
12: T = t + 273.15
13:
14: OUTPUT a, b, c, d

```

/\* 水蒸気の定圧分子熱式の決定 \*/

```

[ 計算結果 ]
a = 4.278842
b = -0.012977
c = 2.382229E-005
d = -1.417749E-008

```

例えば、管摩擦係数はレイノルズ数の値により層流の式と乱流の式を使い分けて計算されるが、このような場合も条件付きの式として簡潔に表現できる。

この他に配列の大きさを調節したり、同じ式を何度も書かなくてすむように式を書く上での柔軟性をもたせるために、マクロやパラメータなどの表記法が用意されている。

以上、言語としての EQUATRAN-M の基本機能を説明したが、付言すれば、微分方程式や積分方程式も離散化すれば連立方程式として表わされるので、これを解くことはできるが直接数値積分をして解を求めるることはできない。今後の課題である。

#### 4. EQUATRAN-M の使い方

次に、パソコン上で EQUATRAN-M をどう使うか、操作法について説明する。途中で操作方法がわからなくなったら「HELP」キーを押せばガイドが表示され、HELP の終了で元の画面に戻る。なお、キーボード配列は PC-9801 シリーズのそれに準拠する。

##### <例題 1.5：温度の単位変換>

摂氏度 [°C] と華氏度 [°F]、ケルビン度 [°K] の単位換算は次式で与えられる。

$$F = 9/5 C + 32 \quad (1-11)$$

$$C + 273.15 = K \quad (1-12)$$

任意の華氏度に対する他の温度を求めよ。

解) EQUATRAN-M では (1-11), (1-12) 式をそのまま書く。

```
/* Temperature Unit Conversion */
F=9/5*C+32
C+273.15=K
INPUT F
```

このリストのように、EQUATRAN-M の記法に従って書かれた方程式、宣言文、注釈の集まりを今後ソーステキストという。

#### 4.1 ソーステキストの入力

EQUATRAN-M を起動すると 図1-1 のオープニング画面が現われるが、スペースキーを押すと図1-2 の基本画面に変る（もし、いずれの画面でもないときはコマンド（最下行）を ‘NEW’ とキーインし回を押せば基本画面となる）。ソーステキストを入力するために、コマンドが A であることを確認して回を押すと図 1-3 のように行番号を示す 1：の 1 文字後にカーソルが移動し、入力の準備完了である。そこで、リストの第 1 行目

```
/* Temperature Unit Conversion */
とキーインし（各行は行番号を示す：の後に空白が 1 文字分あり、その次の位置から 72 文字分がテキスト領域で、どこから始めてもよい）、回するとカーソルは 2 行目に移る。テキストを見易くす
```



図 1-1 オープニング画面




図 1-2 基本画面

るために 1 つ以上のスペースを打って回すれば 2 行目は空白行になり、カーソルは 3 行目に移る。この要領で次々と方程式をキーインし、最終行の N PUT 文をキーインしたら回を 2 度続けて押せばソーステキストの入力は完了する（図1-3）。

#### 4.2 ソーステキストの修正

ここで、画面をもう一度見直してもし誤りを発見したら ↑ ↓ ← → のカーソル移動キーを使ってカーソルを誤り文字の上へ移動させ、正しい文字を打直して訂正し、（何文字でも何行でもよい）、回を押せば修正完了である。このとき、文字の挿入や削除は |INS| や |DEL| キーを使えばよろしい。行を追加したり削除したり、また、行をコピーしたり移動したりすることはコマンドでいとも簡単にできる。画面に表示されるのは 20 行であるが、画面の送り戻し（スクロール）もコマンドで自由自在である。このように画面の中を自在に修正、編集する機能をもったフルスクリーンエディタが提供されているので、ソーステキストの入力、誤りの訂正を気軽に手早くすませることができ、大変便利で作業効率の向上に役立っている。

画面編集に使われる主なコマンドを表1-14 にまとめておく。

#### 4.3 計算の実行

さて、ソーステキストの入力と修正が終ったら計算を実行させるためコマンドに ‘RUN’ とキーインし回を押すと、EQUATRAN-M はコンパイルを行い、書式の不備や方程式の過不足があればエラーメッセージをコマンド行の下に表示する。エラーメッセージが出たらしかるべき訂正を施こし、再度 ‘RUN’ 回する。すると、実行時入力として INPUT 文で指定した変数 F の値を要求してくれる（プロンプトが点滅する）ので、100 とキーインし回すれば直ちに 図1-4 のように全ての変数の値が表示される（OUTPUT 文を省略すると全変数を出力する）。

さて、続けて 60°F に対する C, K の値を求めるには再コンパイルの必要がないから、コマンドに ‘GO’ とキーインし回すると F の値を要求してくれるで 60 とキーインして回すればよい。このように入力値を種々変えてケーススタディすることが容易にできる。

入力したソーステキストをディスクに格納する

表 1-14 画面編集コマンド

機能	コマンド	例
行の追加	A (Append)	A (最終行に追加), A3 (第3行の次に追加)
行の削除	D (Delete)	D4 (第4行を削除), D5-6 (第5～6行を削除)
行の選択*	S (Select)	S5 (第5行を選択), S5-7 (第5～7行を選択)
行のコピー	C (Copy)	C7 (選択された行を第7行の直後にコピー)
行の移動	M (Move)	M6 (選択された行を第6行の直後に移動)
リスト表示	L (List)	L (第1行から表示), L6 (第6行から表示)
スクロール	F (Full)	F (1画面前送り), F- (1画面逆戻し)
〃	H (Half)	H (半画面前送り), H- (半画面逆戻し)

\* SコマンドはCまたはMコマンドと組合せて使用される

```

ソース   I (S)   I (D)   I (O)   I
1: /* Temperature Unit Conversion */
2:
3:   F = 9/5 * C + 32
4:   C + 273.15 = K
5:
6:   input F

コマンド>F

```

図 1-3 ソーステキスト入力

```

ソース   I (S)   I (D)   I (O)   I
/* Temperature Unit Conversion */
[ 入力値(=) または 初期値(#) ]  =
F = 100

[ 計算結果 ]
C = 37.777778
F = 100
K = 310.927778

コマンド>L

```

図 1-4 計算結果

には、コマンドで

SAVE "textname"

とキーインし回す。textname はソーステキストの名前を示し、英字で始まる 8 文字以内の英数字で表す。一方、既に格納されているソーステキストを画面に呼び出すには 'LOAD' 回すすればソーステキスト名の一覧が表示されるので、カーソルを所望のソーステキスト名の上に移動させて回すとそのソーステキストが画面に表示される。この他各種のコマンドが用意されているが、表1-15によく使う基本的なコマンドをまとめておく。

#### 4.4 日本語入力

注釈は/\*と\*/でくくって示され、ソーステ

表 1-15 基本的なコマンド

機能	コマンド*
画面上のソーステキストを消去する	NEW
実行する	RUN
入力値を変えて実行する	GO
ソーステキストをディスクに格納する	SAVE ["textname"]**
ソーステキストをディスクから読み出す	LOAD ["textname"]**
画面のソーステキストにディスクから追加する	MERGE "textname"
ソーステキストを印刷する	TPRINT
計算結果を印刷する	OPTION または COPY

\* コマンドは最初の 2 文字で代表することができる

\*\* [ ] は省略可能を意味する

キストのどこにあってもよい。注釈には日本語（かな、漢字）の使用が許されており、ドキュメントとしての読み易さを随分助けている。日本語の入力方法はパソコンの OS が提供している手順に従う。PC9801 の MS-DOS の下では [CTRL] キーを押しながら [XFER] キーを押せば日本語入力のモードになり、[カナ] をシフトダウンしてからファンクションキー [F1] を押せばカナ・漢字変換が可能な状態になる。漢字を入力したい位置にカーソルを設定しておき、カナで読みをキーインした後 [XFER] を押せば候補となる漢字が表示されるので、カーソル移動キーで目的の漢字を選んで回すればよい。カナを入力するときはカナをキーインして直接回す。また、ひらがなを入力するにはファンクションキー [F3] を押してかなモードに切換える。日本語入力モードから抜けるには、再び [CTRL] + [XFER] すればよい。

以上、EQUATRAN-M の全体像を概観してきたが、パソコンによる技術計算の世界が身近かなものとして広がっていくのがおわかりいただけた

と思う。次回は EQUATRAN-M のより進んだ機能について説明をする。

#### 参考文献

- 1) 三増, 須加, 広瀬: 最近の化学工学, pp. 43~65, 丸善 (1966).
- 2) Miyahara, Sadotomo, Kitamura : *J. Chem. Engrs, Japan* Vol. 3, No. 3, p. 157 (1970).
- 3) 宮原, 佐渡友: 化学工学, Vol. 35, No. 12, p. 1347 (1971).
- 4) 奥, 三増, 羽深: プロセスの制御と計算機制御, pp. 259~291 丸善 (1973).
- 5) 佐渡友, 宮原: 化学工学論文集, Vol. 7, No. 1, p. 33 (1981).
- 6) Oguchi, Okano : Computer-Aided Process Plant Design, pp. 695~710, Gulf Publishing (1982).
- 7) 大島, 正野: *bit* Vol. 9, No. 9, p. 975 (1977).
- 8) Peterson, Chen, Evans : *Chem. Eng'g* June 5 p. 149, 150 & Aug. 28, p. 107, 109 (1978).
- 9) Oguchi, Mitsunaga : International Congress Contribution of Computers to the Devel. of Chem. Eng'g & Ind. Chem., Paris (1978).
- 10) 小口, 横山, 佐藤: 化学工学協会第50年会要旨集 E302 (1985).
- 11) 化学工学協会: 化学工学便覧, 改訂四版, p. 27 (1978).
- 12) 同上 p. 71.
- 13) International Critical Tables Vol. 5, p. 82 McGraw-Hill (1929).
- 14) 矢木, 西村: 化学プロセス工学, p. 155, 丸善 (1969).